Creation of WRKLPPOR

1. Create Display File
A display file is required. The display file in this example will use five fields for the subfile panel, and will use all fields (in LPPOR) in the maintenance/display panel.

The five fields used in the subfile panel will be:

OROBJNAM   -  Object Name

OROBTYP      -  Object Type

ORPRODID    -  Product ID

ORPRODOPT - Product Option

OROBJDES    -  Object Description

Any fields (in the file to be maintained) not explicitly displayed need to be carried within the panel as hidden fields. This means that the display file will contain the input buffer.

2. Create Query Format Files
The ALSP uses the OPNQRYF command to create ODPs for the input files. There will be one query format file for each possible sort order. In our example, there are ten possible sort orders (sequence), one for each field (both ascending and descending). 

Ten memberless files will be created, each one specifying the appropriate key sequence.

3. Create UI Program
There are four places the UI needs to be modified to reflect field names. See the UI documentation.

4. Create Application Logic Service Program
An application logic service program needs to be created. LPPORAL is a possible name. 

The following changes need to be made to an existing ALSP:

4a. Input files

  * * * * * * * * * * * * * * * * * * * * * * * * * * * * *    

  *  Input Files                                                      

  * * * * * * * * * * * * * * * * * * * * * * * * * * * * *    

 flpporl1aif   e   k disk   UsrOpn ReName(lppor:lpp1a)  

 flpporl1dif   e   k disk   UsrOpn ReName(lppor:lpp1d)  

 flpporl2aif   e   k disk   UsrOpn ReName(lppor:lpp2a)  

 flpporl2dif   e   k disk   UsrOpn ReName(lppor:lpp2d)  

 flpporl3aif   e   k disk   UsrOpn ReName(lppor:lpp3a)  

 flpporl3dif   e   k disk   UsrOpn ReName(lppor:lpp3d)  

 flpporl4aif   e   k disk   UsrOpn ReName(lppor:lpp4a)  

 flpporl4dif   e   k disk   UsrOpn ReName(lppor:lpp4d)  

 flpporl5aif   e   k disk   UsrOpn ReName(lppor:lpp5a)  

 flpporl5dif   e   k disk   UsrOpn ReName(lppor:lpp5d)  

These are the files created in step 2. They are used be the OPNQRYF statements.

4b. Update files

 * * * * * * * * * * * * * * * * * * * * * * * * * * * * *       

 *  Update Files                                                        

 * * * * * * * * * * * * * * * * * * * * * * * * * * * * *       

flppor  uf a e     k disk UsrOpn ReName(lppor:updateFile) 

 * * * * * * * * * * * * * * * * * * * * * * * * * * * * *      

We read through logicals, but we update the physical.

4c. Data integrity files.

Any files used to edit a record

4d. /COPYs

The /COPY precompiler commands need to be updated, so that the appropriate header files are included

4e. File buffer

  * * * * * * * * * * * * * * * * * * * * * * * * * * * * *   

  *  file buffer                                                         

  * * * * * * * * * * * * * * * * * * * * * * * * * * * * *       

 d LPPORBFR    E DS                  ExtName(LPPOR)                 

  * * * * * * * * * * * * * * * * * * * * * * * * * * * * *   

  *  copies of file buffer                                               

  * * * * * * * * * * * * * * * * * * * * * * * * * * * * *      

 d                 DS                                                    

 d LPPORHLD                            Like(LPPORBFR)                  

 d savQrySlt                           Like(LPPORBFR)                  

This externally defined data structure will be used as the basis of a shared, pointer based, variable used by all programs requesting services of the ALSP.

4f. User Index Data Structures

It is necessary that we store all selection options in such a manner that we can retrieve them in the same order as the display sequence. Many selection options can be chosen, using the roll keys, position to field and top bottom functions, in an order very different than the displayed sequence. When executing these selection options, the order needs to be the same as the display sequence. It is necessary to define a data structure for each sortable column (ascending/descending doesn’t matter here). The first data structure would look like:

 * * * * * * * * * * * * * * * * * * * * * * * * * * * * *   

 *  User Index Data Structures                                          

 *                                                                      

 *  We don't want to create a user index for each access path, so       

 *  instead, we put all possible key fields in data structures,         

 *  and move from LPPORBFR into QFxDS, creating the different key     

 *  structures. Implications of descending keys discussed later.        

 * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *      

DQF1DS            DS                                                    

D  QF1BLANK                           Like(blank) Inz(' ')              

D  QF1_OBJNAM                         Like(OROBJNAM)                   

D  QF1_OBJTYP                         Like(OROBJTYP)                  

D  QF1_PRDID                          Like(ORPRODID)                      D  QF1_PRDOPT                         Like(ORPRODOPT)                   

D  QF1_OBJDES                         Like(OROBJDES) 
The preceding data structure is used whenever Object Name is the select sort column (ascending or descending). When Object Type is the selected sort column, the data structure is:  

DQF2DS            DS                                                    

D  QF2BLANK                           Like(blank) Inz(' ')              

D  QF2_OBJTYP                         Like(OROBJTYP)                  

D  QF2_OBJNAM                         Like(OROBJNAM)                   D  QF2_PRDID                          Like(ORPRODID)                      D  QF2_PRDOPT                         Like(ORPRODOPT)                   

D  QF2_OBJDES                         Like(OROBJDES) 
and so forth.

4g. Standalone variables. 

A number of standalone variables are initialized with program specific values (e.g. the names of the user space and user index). These will need to be updated.

4h. Constants. 

A number of constants are defined with program specific values (e.g., the various close file constants used as arguments in a call to QCMDEXC).

4i. Public Procedures:

initialize – needs to refer to the appropriate input buffer

retrieve buffer – needs to refer the appropriate input buffer

edit – The heart of the difference between the various ALSPs is the edit procedure. This procedure can not be maintained, or updated; it will have to be created. This makes since, every file has different edit criteria.

update – Copy logic will need to be twitched

 Csr   $Cpy          BegSR                                               

 C                   eval      error = io(WRITE)                         

 C                   Eval      new_OBJNAM    = OROBJNAM                                                                              C                    Eval      new_OBJTYP    = OROBJTYP                    

  * ... delete any existing detail records ...                           

 C                   DoU       ( *IN99 )                                 

 C     newKey        Delete    updateFile                         99     

 C                   EndDo                                               

  * ... copy new detail records ...                                      

 C     oldKey        Chain     LPPOR                           99     

 C                   DoW       ( Not *In99 )                             

 C                   Eval      OROBJNAM    = new_OBJNAM                      

 C                   Eval      OROBJTYP_L  = new_OBJTYP                    

 C                   Write     updateFile                                

 C     oldKey        ReadE     LPPOR                               99 

 C                   EndDo   

 C                   EndSR                                            

set sequence – This procedure uses a passed field name to set sort sequence. These field names will need to be changed to reflect the new file.

load defaults – This procedure loads fields with default values. It is used in conjunction with the add panel. 

4j. Private procedures 

io – This procedure performs all file i/o operations. It contains a KLIST for each sortable column, and all of the various RPG operations against the master file. One of the KLISTs, and the  ReadPE subroutine are shown below:

 UpdateKey     KList                               

               KFld                    OROBJNAM     

               KFld                    OROBJTYP   

 QF1Key        KList                               

               KFld                    OROBJNAM     

               KFld                    OROBJTYP   

               KFld                    ORPRODID     

               KFld                    ORPRODOP   

               KFld                    OROBJDES   
csr   $ReadPE       BegSR                                               

c                   Select                                              

c                   When      ( openFile = QF1A )                       

c     QF1Key        ReadPE    lpporl1a                             99 

c                   When      ( openFile = QF1D )                       

c     QF1Key        ReadPE    lpporl1d                             99 

c                   When      ( openFile = QF2A )                       

c     QF2Key        ReadPE    lpporl2a                             99 

c                   When      ( openFile = QF2D )                       

c     QF2Key        ReadPE    lpporl2d                             99 

c                   When      ( openFile = QF3A )                       

c     QF3Key        ReadPE    lpporl3a                             99 

c                   When      ( openFile = QF3D )                       

c     QF3Key        ReadPE    lpporl3d                             99 

c                   When      ( openFile = QF4A )                       

c     QF4Key        ReadPE    lpporl4a                             99 

c                   When      ( openFile = QF4D )                       

c     QF4Key        ReadPE    lpporl4d                             99 

c                   When      ( openFile = QF5A )                       

c     QF5Key        ReadPE    lpporl5a                             99 

c                   When      ( openFile = QF5D )                       

c     QF5Key        ReadPE    lpporl5d                             99 

c                   EndSL

csr                 EndSR

open view – This procedure builds the select clause, and creates an ODP. It will need to be modified to reflect the new OPNQRYF statements (contained in arrays; these will need to be modified as well).

init – This procedure opens files. It will need to be updated, as in the following:

  *-------------------------------------------------------------------   

 p init            b                                                     

  *-------------------------------------------------------------------   

 d                 pi             5i 0                                   

 C                   Open      lppor                                  

 C                   Open      com0002f2                                 

 c                   Eval      error = 0                                 

 c                   Return    error                                     

  *-------------------------------------------------------------------   

 p init            e                                                     

closeall – This procedure closes files. It will need to be modified in a manner analogous to the init modifications.

lokey, hikey – These procedures load the input buffer with special values, for position to beginning and end of file.

loadds, unloadds – These procedures move fields from the input buffer to their appropriate position within the QFxDS data structure (used for storing selection options). They need to be modified similar to the following:

C                   When      ( openfile = QF1A )                       

C                   Eval      QF1_OBJNAM = OROBJNAM                      

C                   Eval      QF1_OBJTYP = OROBJTYP                    

C                   Eval      QF1_PRDIDO = ORPRODID                      

C                   Eval      QF1_PRDOPT = ORPRODOPT                    

C                   Eval      QF1_OBJDES = OROBJDES 

C                   Eval      QFxDS = QF1DS                             

C                   When      ( openfile = QF1D )                       

c     fromDS:toDS   XLate     OROBJNAM     QF1_OBJNAM                     

c     fromDS:toDS   XLate     OROBJTYP     QF1_OBJTYP                   

c     fromDS:toDS   XLate     ORPRODID     QF1_PRDID                     

c     fromDS:toDS   XLate     ORPRODOPT    QF1_PRDOPT                   

c     fromDS:toDS   XLate     OROBJDES     QF1_OBJDES                   

C                   Eval      QFxDS = QF1DS                             

descendkey – When using position to logic with a descending key, the position to value must be twitched. Rather than being right padded with blanks (x’40’), it should be right padded with *HIVALs (x’FF’), otherwise the cursors setting operations will position after the record, not before it.

closeview – Used to close ODPs prior to openview. Needs to be modified to refer to the correct files.

Notes: Adding a function key to the UI. Following is the code that was added to the SQLQRYMON program when the refresh functionality was included (the added code is highlit):

 * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *  

 *                                                                      

 * ... process command function keys ...                                

 *                                                                      

 * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *  

c     $SPrcssFnc    BegSR                                               

 * ... validate function key authority ...                              

C                   Eval      msgID = fncSecure(msgpgm:user:keyPressed) 

C                   If        ( msgID  = *Blanks )                      

c                   Select                                              

c                   When      ( keyPressed = PagForWrd )                

c                   Eval      state = SPagFwd                           

c                   When      ( keyPressed = PagBakWrd )                

c                   Eval      state = SPagBak                           

C                   When      ( keyPressed = F06 )                      

c                   ExSR      $WAddState                                

C                   When      ( keyPressed = F07 )                      

c                   Callp     clearMsgs(MsgPgm)                         

C                   Call      'QCAEXEC'                            99   

C                   Parm      SbmFilBld     P11             512         

C                   Parm      512           P12              15 5       

C                   If        ( *In99 )                                 

c                   Callp     clearMsgs(MsgPgm)                         

C                   Eval      error = sndMsg(AQM7030:                   

c                                            SQLQRYMON:                 

c                                            msgSubDta:                 

C                                            msgPgm)                    

C                   EndIf                                               

c                   Eval      state = SReDisp                          

c                   When      ( keyPressed = F13       )               

c                   Eval      state = SRepeat                          

c                   When      ( keyPressed = F14       )               

c                   Eval      state = SSort                            

c                   When      ( keyPressed = F17       )               

c                   Eval      state = SFrstPag                         

c                   When      ( keyPressed = F18       )               

c                   Eval      state = SLastPag                         

c                   When      ( keyPressed = Enter     )               

c                   Eval      state = SEnter                           

C                   Other                                        

C                   Eval      error = sndMsg(AQM7003:            

c                                            SQLQRYMON:          

c                                            msgSubDta:          

C                                            msgPgm)             

c                   Eval      state = SExFmt                     

c                   EndSl                                        

C                   Else                                         

C                   Eval      erropt = *On                       

C                   Eval      error = sndMsg(msgID:              

c                                            SQLQRYMON:          

c                                            msgSubDta:          

C                                            msgPgm)             

c                   Eval      state = SexFmt

C                   EndIf                                               

c                   EndSR                                                                    
Adding a function code is very simple. If the function code is handled within the UI (e.g. PagBakWrd), a state subroutine is created, and the state variable loaded with the appropriate state. If the function code is handled outside of the UI, then all that is needed is a program call. Note that unsuccessful completion of the called program is monitored, and an appropriate messagre sent if circumstances warrant.
